Supplemental Material

Definition of Crude referral proportions: Overall crude referral ratios were defined as the total number of referrals divided by the total number of incident ESRD patients in a year. A 4 year crude facility performance measure, facility-specific referral proportions, was calculated as the number of referrals over the study years per incident patients within the facility.

Determining Days at risk for the Cox Proportional Hazard Model: A patient-record began each time the patient was determined to be at a different facility within the first year of ESRD incidence. For example, a patient who was in the same facility for the entire year will result in one patient record, however, when a patient received services at different facilities during the year, the days at risk at the facilities generated one patient record per facility for that year. End of follow-up for each patient/facility for that year was defined as either the day of the event (referral), transplantation, death or the 365^{th} day after ESRD incidence if the event had not yet occurred. Time at risk for each record was calculated as the fraction of the year the patient was in a facility until end of follow-up.

Details of Cox Proportional Hazards Model and STReR computation: A 2-stage Cox proportional hazards (CPH) model was used to compute the expected number of first-time referrals at a facility. Each patient/facility record included a censor indicator of whether a referral was made within that period. Patients were censored for transplant, death or end of 1 year follow up without referral. Stage 1 of the CPH model was stratified by facility and adjusted for patient case-mix. Stratifying by facility ensured that facility-level heterogeneity is accounted for in the baseline survival probabilities. Separate incident year intercepts were included to test for time effects. The predicted values from the stage 1 Cox model were used as offset term in the stage 2 of CPH model, which estimated a common baseline hazard function and included an adjustment for the covariate-adjusted population referral rates. Expected number of referrals per person at each facility were obtained by aggregating the Martingale residuals from stage 2 of the CPH model. The STReRs were obtained as a ratio of observed to expected referrals at the facility level.

Definition of evaluation, waitlisting and transplantation proportions: The evaluation, waitlisting and transplantation proportions was calculated at the facility level and were defined as the number of events within 6, 12 and 24 months respectively over all referred patients in that facility.

Details of Reliability calculation: An inter-unit reliability (IUR) measure was computed for each year using a bootstrap approach. Because STReR is not a simple average across patients in a facility, a resampling scheme was used to estimate the within facility variation (19). Based on 1000 bootstrap samples the proportion of measure variability attributable to the between-facility variance was computed.

SAS code for STReR calculation:

```
/*** define study cohort***/
data one; set radiant.usrds_referral_facility_1;
      by USRDS_ID FIRST_SE;
      format esrddate MMDDYY10.;

      esrddate=FIRST_SE;
      followup_1yr=intnx('day', FIRST_SE, 365);
      inc_year=YEAR(esrddate);
      followup_end= min(died, followup_1yr, tx1date);
      format  followup_1yr followup_end mmddyy10.;
      if (FIRST.USRDS_ID) AND  (FIRST_SE ne .) AND
      (MDY(1,1,2008) <=FIRST_SE <=MDY(12,31,2011)) AND ((referral_date >=
FIRST_SE) or referral_date=.) ;

run;

/***get treatment history of patients; record facility transitions***/
data Phistory; set radiant.Treatment_history;
data Phistory_long; set Phistory;
array bdate{11} begdate1-begdate11;
array endate{11} enddate1-enddate11;
array rx{11} $ rxgroup1-rxgroup11;
array fac{11} provusrd1-provusrd11;
do time=1 to 11;
 begdate=bdate{time};
 enddate=endate{time};
 rxgrp=rx{time};
 facility=fac{time};
 if (facility ne .)  and (begdate ne .) and (rxgrp in
('1','2','3','5','7','9')) then output;
 end;
 format begdate mmddyy10. enddate mmddyy10.;
 keep USRDS_ID begdate enddate rxgrp time facility;
 run;

/***Merge patient history with the study cohort****/

proc sql;
  create table two as
  select  *
  from one, Phistory_long
  where one.USRDS_ID=Phistory_long.USRDS_ID
  order by one.USRDS_ID, Phistory_long.begdate, Phistory_long.enddate;
quit;

/***set the start and followup end date /patient/facility***/
data three; set two;
by USRDS_ID;
retain startbegdate;

if first.USRDS_ID then do;
tempbegdate=begdate;
startbegdate=begdate;
end;
```

```
if facility = lag1(facility) then tempbegdate=startbegdate;
else do ;
tempbegdate=begdate;
startbegdate=tempbegdate;
end;

timediff=DATDIF(begdate,tempbegdate, 'ACT/ACT');
analysis_start= MAX(esrddate, tempbegdate);
start_year=YEAR(analysis_start);

followup_end_indfac= min(died, followup_end, tx1date, referral_date,
enddate);

format analysis_start MMDDYY10. tempbegdate  startbegdate mmddyy10.
followup_end_indfac MMDDYY10. ;

pdays=DATDIF(analysis_start, followup_end_indfac, 'ACT/ACT');
if (pdays > 0) then output;
run;


/****keep unique facilities/patients****/

proc sort data=three out=foursorted;
by USRDS_ID  facility analysis_start  descending followup_end_indfac;

run;
proc sort data=foursorted out=four nodupkey;
by USRDS_ID  facility analysis_start ;
run;

/**calculate number of patients/facility and apply >=5 incident patients/year rule*/
proc sql;
create table patsum as
   select  facility,inc_year, count(distinct USRDS_ID) as numpatients
  from four
 group by facility, inc_year;
quit;
proc univariate data=patsum;
class inc_year;
 var numpatients;
 run;

proc sort data=four; by facility inc_year; run;
data four_up; merge four patsum;
by facility inc_year;
run;
data five; set four_up;
if numpatients >=5;
run;


/*****define the risk-adjustment factors***/

data temp; set five;
ind_ref=0;/**indicator of referral in each interval***/
if referral_date ne . then do;
```

```
if (referral_date >= analysis_start) and (referral_date <=
followup_end_indfac) then ind_ref=1;
end;
agenum=inc_age;
if agenum ne . then do;
if 18<agenum<=30 then agecat2=1;
else if 30<agenum<=40 then agecat2=2;
else if 40<agenum<=50 then agecat2=3;
else if 50<agenum<=60 then agecat2=4;
else agecat2=5;
end;
else agecat2=99;


yr2008=(inc_year=2008);
yr2009=(inc_year=2009);
yr2010=(inc_year=2010);
yr2011=(inc_year=2011);


   if race=3 then race_cat='Black';
   else if race=4 then race_cat='White';
   else race_cat='Other';


                  *Ischemic heart disease, cardiac arrest, and MI are all
                        categorized as atherosclerotic heart disease in 2005
                        med evidence form;

      if carfail ne ' ' or como_chf ne ' ' then do;

          if carfail = 'Y' or como_chf = 'Y' then
                  chf = 1;
            else chf = 0;
            end;
            else chf=99;


          if cararr ne ' ' or mi ne ' ' or ihd ne ' ' or como_ashd ne ' '
then do;
            if cararr = 'Y' or mi = 'Y' or ihd = 'Y' or como_ashd = 'Y' then
                  ashd_new = 1;
            else ashd_new = 0;
        end;
        else ashd_new=99;

            if como_copd ne ' ' or pulmon ne ' ' then do;
            if como_copd = 'Y' or pulmon = 'Y' then copd_new = 1;
                  else copd_new = 0;
            end;
            else copd_new=99;

        if como_othcard ne ' ' then do;
            if como_othcard = 'Y' then como_othcardnew = 1;
                  else como_othcardnew = 0;
            end;
            else como_othcardnew = 99;
```

```
          if pvasc ne ' ' or como_pvd ne ' ' then do;
          if pvasc = 'Y' or como_pvd = 'Y' then pvasc_new = 1;
                    else pvasc_new = 0;
               end;
               else pvasc_new = 99;


               if cva ne ' ' or COMO_CVATIA ne ' ' then do;
          if cva = 'Y' or COMO_CVATIA = 'Y' then cva_new = 1;
                    else cva_new = 0;
               end;
               else cva_new=99;

             *Combine cardiac diseases (and other cardiac) to create an
indicator variable of CVD;
             if ashd_new ne 99 or pvasc_new ne 99 or cva_new ne 99 or copd_new
ne 99 or COMO_OTHCARD ne ' ' then do;
             if ashd_new = 1 or pvasc_new = 1 or cva_new = 1 or copd_new = 1
or COMO_OTHCARD='Y'
                    then CVD = 1;
                    else CVD = 0;
             end;
           else CVD = 99;

         if hyper ne ' ' or como_htn ne ' ' then do;
         if hyper = 'Y' or como_htn = 'Y' then hypertension = 1;
                    else hypertension = 0;
             end;
             else hypertension = 99;

         if diabins ne ' ' or COMO_DM_INS ne ' ' or COMO_DM_NOMEDS ne ' ' or
COMO_DM_ORAL ne ' ' or COMO_DM_RET ne ' ' then do;
             if diabins = 'Y' or COMO_DM_INS = 'Y' or COMO_DM_NOMEDS = 'Y'
                    or COMO_DM_ORAL = 'Y' or COMO_DM_RET = 'Y' then diabetes =
1;
                    else diabetes = 0;
             end;
             else diabetes=99;


             if smoke ne ' ' or COMO_TOBAC ne ' ' then do;
             if smoke = 'Y' or COMO_TOBAC = 'Y' then smoke_new = 1;
                    else smoke_new = 0;
             end;
             else smoke_new = 99;

          if drug ne ' ' or como_drug ne ' ' then do;
                    if drug = 'Y' or como_drug = 'Y' then drug_new = 1;
                          else drug_new = 0;
               end;
               else drug_new=99;

          if cancer ne ' ' or como_canc ne ' ' then do;
          if cancer = 'Y' or como_canc='Y' then cancer_new=1;
                    else cancer_new=0;
```

```
             end;
             else cancer_new=99;


      **Impossible BMI values - set to missing;
            if bmi > 60  then bmi = .;
                  else bmi=bmi;

            ** BMI > 35;
          if bmi ne . then do;
            if bmi >= 35 then bmi_35 = 1;
            else bmi_35 = 0;
          end;
           else bmi_35=99;

       ***history of nursing home status;

            if como_inst_nurs=' ' then hist_nursing_home= 'MISS';
            else hist_nursing_home= como_inst_nurs;

run;
```

/*****Calculate STReR: stages 1-2 of the Cox PH model ***/

```
proc phreg data=temp;
class race_cat(ref='White')  male(ref='0')  agecat2(ref='3') bmi_35(ref='0')
 chf(ref='0') ashd_new(ref='0')  como_othcardnew(ref='0') cva_new(ref='0')
pvasc_new(ref='0')  copd_new(ref='0') hypertension (ref='0') diabetes
(ref='0') cancer_new(ref='0')  smoke_new (ref='0')
hist_nursing_home(ref='N');
 strata facility;
 model pdays*ind_ref(0)= agecat2 race_cat male yr2008 yr2009 yr2010 bmi_35
chf ashd_new como_othcardnew cva_new pvasc_new copd_new hypertension diabetes
smoke_new cancer_new hist_nursing_home/ties=breslow rl;
id USRDS_ID inc_year;
 output out=stage1_srr resmart=marting  resdev=rdev xbeta= xbeta_srr;
 run;

proc  phreg data=stage1_srr noprint;
model pdays*ind_ref(0)=/offset=xbeta_srr ties=breslow;
id USRDS_ID facility inc_year;
output out=srr resmart=res_srr;
run;

data srr_2 (keep=USRDS_ID inc_year facility expectref res_srr pdays ind_ref
);
set srr;
expectref=ind_ref-res_srr;
run;


/***calculate facility STReR****/
proc sql;
create table srr_out as
  select  facility, count(unique USRDS_ID) as patients, sum(ind_ref) as
obs_ref, sum(expectref) as expected_ref, sum(ind_ref)/sum(expectref) as STReR
  from srr_2
```

```
  group by facility;
quit;

/****calculate exact confidence intervals for STReRs***/

data srr_out; set srr_out;

FORMAT LCL UCL 8.2;
IF obs_ref=0 THEN LCL=0;
IF obs_ref>0 THEN
LCL=(quantile('CHISQ',.025,2*obs_ref)*0.5)/expected_ref;
UCL=(quantile('CHISQ',.975,2*(obs_ref+1))*0.5)/expected_ref;

cont=0;
if (LCL < 1) AND (UCL >= 1) then cont=1;
else if (LCL > 1) then cont=2;

keep facility patients obs_ref expected_ref log_exp STReR LCL UCL cont;
run;
```

Supplementary Table A1. Stage 1 Cox Proportional Hazard model estimates of patient case-mix factors on referral for transplantation within 2-year follow up period.

Parameters	Referent Group	Coefficient	Standard error	Hazard Ratio (95% CI)
Age group				
19-30		0.38	0.08	1.46 (1.25, 1.71)
31-40		0.33	0.06	1.39 (1.23, 1.56)
51-60		-0.20	0.05	0.82 (0.74, 0.91)
60+	41-50	-0.73	0.06	0.48 (0.43, 0.54)
Race group				
African-American	White	0.11	0.05	1.12 (1.02, 1.23)
Other		0.23	0.13	1.26 (0.98, 1.64)
Male	female	0.13	0.04	1.14 (1.06, 1.23)
Incident ESRD Year				
2008		-0.26	0.06	0.77 (0.69, 0.86)
2009		-0.16	0.06	0.85 (0.76, 0.95)
2010	2011	-0.04	0.05	0.97 (0.87, 1.07)
BMI $\geq$ 35 kg/m^2	BMI < 35 kg/m^2	-0.02	0.04	0.98 (0.90, 1.07)
Comorbidities				
Congestive heart failure	None	-0.15	0.05	0.86 (0.78, 0.94)
Atherosclerotic heart disease	None	0.03	0.08	1.03 (0.89, 1.21)
Other cardiac disease	None	-0.14	0.07	0.87 (0.76, 0.99)
Cerebrovascular disease	None	-0.31	0.08	0.73 (0.62, 0.86)
Peripheral vascular disease	None	-0.20	0.08	0.82 (0.69, 0.97)
COPD	None	-0.32	0.10	0.72 (0.59, 0.89)
Hypertension	None	0.18	0.06	1. 20 (1.06, 1.36)
Diabetes	None	-0.03	0.04	0.97 (0.90, 1.05)
Tobacco use	None	-0.22	0.07	0.80 (0.69, 0.92)
Cancer	None	-0.59	0.12	0.56 (0.44, 0.70)
Nursing home status	None	-1.63	0.23	0.20 (0.13, 0.31)

Supplementary Table A2: Stage 1 Cox Proportional Hazard model estimates of insurance status on referral for transplantation

A sensitivity analysis was performed to compare the STReRs obtained after risk adjusting for insurance type at start of dialysis. While patients insured by employer groups were more likely to be referred compared to patients with Medicare, the other insurance categories (e.g. Medicaid, other coverage and no coverage) were not statistically significantly different from Medicare (estimates shown below) in this cohort. The results were very similar for the original casemix variables and the facility STReRs calculated w/ and w/o insurance type as a risk adjustment factor had a correlation of 0.98, indicating high agreement.

Parameters	**Referent Group**	**Coefficient**	**Standard error**	**Hazard Ratio (95% CI)**
Insurance status				
Medicaid	Medicare	-0.13	0.07	0.88 (0.76, 1.01)
Employer group		0.57	0.06	1.76 (1.55, 2.00)
Other coverage		0.19	0.11	1.21 (0.97, 1.51)
No coverage		-0.05	0.07	0.95 (0.82, 1.09)

Supplementary Table A3: Facility performance for small facilities (<5 incidents patients/ year)

A sensitivity analysis was performed to compare the STReR measure based on all 321 GA dialysis facilities and those excluding the facilities with <5 incident patients/year (N=72). The results are almost identical in terms of model estimates, however, the STReR estimates for the small facilities, especially for those with 0 observed referral, had very high uncertainty resulting in very wide confidence intervals. Among the 72 excluded facilities, 67(94%) were as expected, and 1 in each better/worse than expected categories (please refer to the table below). The STReR measure could not be calculated for 3 (4.2%) facilities due to lack of variability in the data. The classification of the bigger facilities were almost identical except for small differences in the better/worse than expected category.

Facility size	Better than expected	As expected	Worse than expected
≥5 incident patients/year (N=249)	27 (11%)	191 (77%)	31(12%)
<5 incident patients/year (N=72)	1 (1%)	67 (93%)	1 (1%)

Supplementary Table A4. Characteristics of facilities that performed better than expected, as expected and worse than expected based on the STReR measure

Facility characteristics	Better than expected (N=27)	As expected (N=192)	Worse than expected (N=30)	P-value
For-profit status, n (%)	26 (96%)	154 (82%)	19 (63%)	0.01
Urban, n (%)	23 (92%)	152 (83%)	23 (77%)	0.32
Freestanding, n (%)	27 (100%)	185 (96%)	25 (83%)	0.01
Distance from nearest Tx center (miles): mean(sd)	34.1 (32)	50.6 (48)	99.5 (79)	<0.01
Average # of patients in a year: mean (sd)	68 (15)	62 (23)	70 (15)	0.14